
ABSTRACT
To achieve security certification according to the highest lev-
els of assurance, formal models and proofs of security prop-
erties are required. In the MILS context, this includes for-
malisation of key components – such as separation kernels
– and the formalisation of applications built on top of these
verified components. In this paper, we use the Isabelle/HOL
proof assistant to formalise the Firewall application built on
top of a verified separation kernel according to the model
of Greve, Wilding, and Vanfleet (GWV). This Firewall ap-
plication has been formalised twice after the original effort
by GWV. These different efforts have been compared and
discussed on paper. Our main contribution is to provide a
formal comparison between these formalisations in the for-
mal logic of a proof assistant.

1. INTRODUCTION
To achieve security certification at the highest levels of as-
surance (e.g. EAL6 or EAL7 of the Common Criteria), for-
mal models and proofs are required. In the context of MILS
architectures, this not only means the formalisation of key
components, like separation kernels, but also the formalisa-
tion of more mundane applications and their composition in
a complete system.

Within the EURO-MILS project, we aim at providing a
modelling and validation environment based on the formal-
isation of a generic MILS architecture. This environment
should ease the development of formal models and proofs of
systems built according to the MILS architectural paradigm.
We present and discuss three existing efforts about the for-
mal verification of an application built on top of a veri-
fied separation kernel. This application is a Firewall origi-
nally proposed and formalised by Greve, Wilding, and Van-
fleet [5], who also proved its correctness using the ACL2 the-
orem prover [6]. This formalisation was later replicated by
Rushby [9], who proved the relevant properties in the logic
of the PVS [8] proof system; to do so, he also refined the
axiomatisation of the Firewall behaviour. Subsequently, a

further refinement of this formalisation was proposed by Van
der Meyden [2], who also proves relations between the three
efforts using informal pen-and-paper proofs. Our main con-
tribution is to formalise Van der Meyden’s axiomatisation
and proofs in the Isabelle/HOL proof assistant [7]. We also
re-formulate in Isabelle/HOL the formalisations by GWV
and Rushby and the relations between the three axiomati-
sations. As part of this effort we found a small flaw in Van
der Meyden’s axiomatisation, for which we present a cor-
rected version; we regard this as a confirmation of the value
of the formalisation of mathematics in the logic of computer
proof systems.

In the next three sections, we introduce the Firewall example
application, the GWV model of formalised security, and ex-
press the Firewall in term of the GWV model. Afterwards,
we compare the three different axiomatisations of the Fire-
wall, proving relevant relations between them. We point out
a flaw in the axiomatisation of Van der Meyden, and present
a corrected version. Finally, we formally show how all three
axiomatisations are sufficient to prove the desired properties
of the larger system containing the Firewall, which we take
as the compositional overall correctness proof.

2. THE FIREWALL APPLICATION
The application Greve, Wilding, and Vanfleet used as an
example of their formalisation of security is one that sani-
tises useful but sensitive information for use by an untrusted
application. This so-called Firewall takes as input informa-
tion presented by trusted parts of the system, which may
be be sensitive. It then filters and censors this information
to produce a version that can safely be passed to applica-
tions that are not trusted to handle it securely, and deliv-
ers this sanitised information to a location where the un-
trusted application can find it. Under the assumption that
the Firewall application is the only source of information to
the untrusted application, this should provably ensure that
no sensitive information ever ends up within reach of the
untrusted application.

The Firewall application does not exist in a vacuum. It
runs on top of an operating system of some sort, specified
in more detail in the next section, which provides controlled
access to memory. Its job is to divide the system memory
into segments and enforce limits on which applications can
access which memory segments. To ensure security, it is
assumed that the operating system is configured in such a
way that the Firewall application is the only component that



Figure 1: The Firewall MILS Example.

can write to memory segments that are accessible to the
untrusted application; moreover, it can only write to a single
such a segment, denoted as outbox. This configuration is
depicted in Figure 1.

For the purpose of the Firewall example, Greve, Wilding and
Vanfleet do not try to express in detail what information is
and is not sensitive. Instead, they assume the presence of
a predicate black that, for a given memory segment and
system state, expresses whether or not that segment con-
tains only nonsensitive information. Thus, in a given state,
a memory segment is black if and only if its contents are
not sensitive. The function of the Firewall, then, is to make
sure it only ever writes black data to outbox. The security
requirement we seek to formalise can then be expressed as
requiring that none of the memory segments accessible to
the untrusted application ever becomes nonblack. The main
goal of the different formalisations presented in the remain-
der of this paper is to prove that this is the case under the
assumption that the operating system and Firewall work as
specified.

3. THE GWV MODEL OF SEPARATION
The system model proposed by Greve, Wilding and Van-
fleet [5] (GWV) guarantees a security property called Sep-
aration. Extensions and variations of this model have then
been proposed [10, 4] and discussed [3, 1]. We will nonethe-
less use the original GWV model [5], which is sufficient for
the purposes of this paper.

The GWV model defines a mathematical formulation of sys-
tems similar to the one presented in Figure 1. At its base, a
GWV system is a deterministic state machine, with states
denoted s or t. Execution consists of repeatedly changing
from a state s to the next state denoted next(s). A GWV
system contains a finite set of memory segments, which in
each state s have contents denoted select(s, a) for segment
a. Furthermore, it contains a finite set of partitions, which
represent independent subcomponents akin to processes in
general-purpose operating systems. In any state, a single
partition is currently active and executing; this partition is
denoted current(s) for state s. This basic model is for-
malised in Isabelle parlance using the following axioms:

fixes current :: State⇒ Partition
fixes select :: State⇒ Segment::finite⇒ Value

fixes next :: State⇒ State

Here, State, Partition, Segment, and Value represent arbi-
trary sets. The finiteness condition on Partition is irrelevant
for the correctness of any of the formalised proofs and has
been omitted. Finiteness of Segment, on the other hand,
turns out to be crucial.

The GWV model assumes that the different partitions run
on top of a separation kernel, a basic operating system tasked
with the duty of restricting memory access of partitions to
those accesses that satisfy a given security policy. One part
of this security policy is a set of memory segments segs(p)
for each partition p describing the memory segments that
that partition is allowed to access. A more subtle compo-
nent of the security policy is an information flow policy,
represented by a binary relation between memory segments,
with the semantics that any computation step that writes
to memory segment a may only do so while reading from a
limited set of input memory segments. Thus, information
may only flow along the edges of the directed graph repre-
sented by the information flow policy. GWV formalise the
information flow policy as a function dia, short for direct
interaction allowed, which for each memory segment a re-
turns a set dia(a) of memory segments that are allowed to
directly influence it:

fixes segs :: Partition⇒ P(Segment)
fixes dia :: Segment⇒ P(Segment)

The security requirement enforced by the separation kernel
can then be expressed as requiring that all memory accesses
must respect the security policy. Greve, Wilding, and Van-
fleet call this property separation. Rather than describing
how a separation kernel might enforce such a security pol-
icy, GWV define separation as requiring that all actions per-
formed by a partition must be independent of the contents
of memory segments that are not allowed to influence the
action. Specifically, they require that whenever a partition
writes to a memory segment a, the contents written may
depend only on the the contents of the segments that are
both in the accessible segments of the executing partition
and dia(a):

assumes Separation: ∀s, t ∈ State, a ∈ Segment.

current(s) = current(t) ∧
select(s, a) = select(t, a) ∧

∀b ∈ dia(a) ∩ segs(current(s)). select(s, b) = select(t, b)

→ select(next(s), a) = select(next(t), a)

This definition reads that for any segment a, for any states
for which both the contents of a and the active partition are
equal, the contents of a in the next state must be a func-
tion of the contents of the memory segments that are both
readable by the active partition and allowed to influence a.
Thus, in changing the contents of a, the executing partition
may not make use of any information other that that allowed
by the security policy.

In the GWV system model, the presence and correct func-
tioning of a separation kernel is taken as an assumption, as
formalised by the Separation axiom. Greve, Wilding, and
Vanfleet propose that this axiom is a useful base for proving
security properties of larger systems that rely on a separa-
tion kernel as a key component. They use their Firewall



application as an example of how to prove security proper-
ties of a larger system by relying on the separation kernel as
a provider of the base security infrastructure.

4. FIREWALL IN GWV
In this section, we formally define both the Firewall appli-
cation and the security property it is supposed to provably
establish.

The Firewall application is a partition F that collects sen-
sitive information from unspecified locations in the system,
and passes a sanitised version of this information along to
a different partition, B, that cannot be trusted to handle it
securely. Relying on the separation kernel to ensure that no
other partitions can write to memory segments accessible to
the untrusted application, this should ensure that the un-
trusted application can only ever get access to information
that has been judged safe by the Firewall.

GWV satisfy this information flow property as the specific
requirement that there is a single memory segment outbox
accessible to B which may be influenced by segments ac-
cessible to partitions other than B. Furthermore, any such
segments that can influence outbox can only be accessible
by F and B. Formally:

fixes B :: Partition
fixes F :: Partition
fixes outbox :: Segment

assumes FW Pol: ∀a, b ∈ Segment, P ∈ Partition.
a ∈ segs(B) ∧
b ∈ dia(a) ∧
b ∈ segs(P ) ∧
P 6= B →
(P = F ∧ a = outbox)

Together with the Separation axiom described in the previ-
ous section, this should suffice to ensure that the only infor-
mation that ends up in segments accessible to B is informa-
tion that the Firewall put there.

As described in Section 2, the behaviour of the Firewall is
modelled using the black predicate, which models the dis-
tinction between sensitive and nonsensitive information. A
memory segment is black in a given state if the contents
of that segment in that state does not contain any sensi-
tive information. The security functionality of the Firewall,
then, is that it never writes any information to outbox that
would cause it to become nonblack. This can be formalised
as the proposition that outbox never changes from black to
nonblack while the Firewall partition F is executing:

fixes black :: State⇒ Segment⇒ B

assumes FW Blackens: ∀s ∈ State.
current(s) = F ∧ black(s,outbox)→
black(next(s),outbox)

We can now state a formal definition of the correctness of
the Firewall application. The desired security property of
the complete system including the Firewall, the untrusted
application, and any possible other applications is that the
segments accessible to B never become nonblack. The re-
quirement that the segments of B start black is not a prop-
erty of the Firewall; the weaker property that can be guar-
anteed by the Firewall is that the segments of B are already

black, they will remain black. Introducing a function run
to express the execution of a number of computation steps,
this can be formalised as follows:

fun run :: N⇒ State⇒ State where

run(0, s) = s
run(Suc(n), s) = run(n,next(s))

theorem FW Correct: ∀s ∈ State, n ∈ N, a ∈ segs(B).
black(s, a)→ black(run(s, n), a)

The combination of the Separation, FW Blackens, and
FW Pol axioms is insufficient to prove the FW Correct
property. For this to be the case, we need further prop-
erties describing the behaviour of the black predicate; for
example, if black data in the segments accessible to B could
become nonblack on its own accord, the FW Correct secu-
rity property quickly falls apart. It is in the axiomatisation
of the black predicate that GWV, Rushby, and Van der
Meyden differ in their approaches. These three approaches
will be the topic of the next section.

5. AXIOMATISATISING BLACKNESS
The behaviour of the Firewall is defined in terms of the
black predicate, which models the property of a segment
of memory of not containing any sensitive information. It
would be expected for this property to satisfy certain ax-
ioms, such as the proposition that a segment cannot change
from black to nonblack without the segment contents be-
ing modified. Certainly, if a nonsensitive chunk of memory
were suddenly to become sensitive without anyone touching
that piece of memory, this would violate our assumptions on
what sensitivity is supposed to mean.

In their publications, GWV, Rushby, and Van der Meyden
take different approaches in characterising the assumed be-
haviour of the black predicate. The basic notion of all three
approaches is that nonblack data cannot be generated from
black data; any computational process that takes only non-
sensitive data as its input must surely produce output that
is also nonsensitive. In this section, we compare the three
different axiomatisations of this notion, and prove relevant
relations between them.

5.1 The GWV Formalisation
One of the main properties that GWV require the black
predicate to have is that in a system in which all segments
are black, all segments will remain black; this is a special
case of the “no spontaneous generation of nonblack data”
principle described above. Another property they require
is that blackness is a function of the content of a memory
segment; it is not allowed that the same data is considered
black or nonblack depending on the context, as this would
allow sensitive data to leak into a completely idle partition.

assumes S5: (∀a ∈ Segment.black(s, a))→
(∀a ∈ Segment.black(next(s), a))

assumes S6: select(s, a) = select(t, a)→
black(s, a) = black(t, a)

These two properties are not sufficient to prove all desired
properties of blackness, however. In particular, we would
like to be able to prove a version of S5 restricted to a par-
ticular partition: the proposition that when all segments of
a partition P are black in a state in which P is active, then



all these segments will still be black in the next state. A
lemma like this has an obvious role to play in any potential
proof of FW Correct.

To make this possible, GWV assume that for every state s
and any segment a, a state can be constructed in which a is
black but which is otherwise identical to s; such a state could
be constructed by, say, wiping the contents of the memory
segment a. To formalise this notion, they posit the existence
of a function scrub producing such a state scrub(a, s), with
straightforward properties:

fixes scrub :: Segment⇒ State⇒ State

assumes S1:
scrub(a, scrub(b, s)) = scrub(b, scrub(a, s))

assumes S2:
a 6= b→ select(scrub(b, s), a) = select(s, a)

assumes S3:
black(scrub(b, s), a)↔ (a = b ∨ black(s, a))

assumes S4:
current(scrub(a, s)) = current(s)

Axioms S1, S2, and S4 together specify that scrub does
not change anything relevant about a state other than the
contents of the scrubbed segment; axiom S3 specifies that a
scrubbed segment is black.

With these properties, the lemma sketched above can be
proven. For if all segments accessible to partition P are
black in a state s with current(s) = P , we can construct a
state t in which all segments are black by scrubbing all other
segments; per axiom S5, the next state next(t) of t also has
all P -accessible states black. But s and t have the same
contents of all memory segments accessible to P ; according
to the Separation axiom, the same must hold for next(s) and
next(t). Thus, per S6, because all P -accessible segments in
next(t) are black, the same must hold for next(s).

The axioms S1 . . . S6 together with the FW Blackens,
FW Pol, and Separation properties are sufficient to prove
the desired FW Correct theorem. We shall prove this by
showing that the GWV axioms are stronger than the Rushby
axioms, and that the Rushby axioms are sufficient to prove
FW Correct, both claims of which are described in the
section below.

5.2 Rushby’s Version
The formalisation proposed by Rushby is a reasonably mi-
nor refinement of the original by GWV. Like GWV, Rushby
includes the two main properties from the GWV formalisa-
tion:

assumes B4: select(s, a) = select(t, a)→
black(s, a) = black(t, a)

assumes B5: (∀a ∈ Segment.black(s, a))→
(∀a ∈ Segment.black(next(s), a))

The difference between the two formalisations is that instead
of a function scrub that replaces the contents of a single
segment by a blackened version, Rushby posits a function
blacken that for a given state scrubs all segments that are
not black.

fixes blacken :: State⇒ State

assumes B1:
black(blacken(s), a)

assumes B2:
black(s, a)→ select(s, a) = select(blacken(s), a)

assumes B3:
current(s) = current(blacken(s))

It is clear that this axiomatisation is quite similar to GWV’s
original. The lemma that was proven for the GWV formal-
isation can be proven for Rushby’s version in a very similar
way. Furthermore, it is clear that the Rushby axioms fol-
low from the GWV axioms: the blacken function can be
constructed by calling scrub on each segment that is not
black1, and the resulting function clearly satisfies the B1,
B2, and B3 axioms.

Unfortunately, formalising this fact in Isabelle proved chal-
lenging. In the logic of Isabelle, the fact that Segment is
finite is expressed as a nonconstructive assertion that a func-
tion f :: N ⇒ Segment and a natural number n exist such
that the segments f(m) for 0 ≤ m < n exactly cover Seg-
ment. Because this is a nonconstructive assertion, we can
only similarly prove in a nonconstructive way that a function
blacken must exist that behaves like a repeated application
of scrub. Both proving that this function exists, and work-
ing with this function to prove properties such as B1 . . . B3
about it, are technically challenging; moreover, they result
in cumbersome proofs that are difficult to understand due
to the technical tricks intermixing the substance of the ar-
gument. We consider this a weakness of the Isabelle proof
system; a more readable way of dealing with nonconstruc-
tively existing entities would be a welcome improvement.
Moreover, to avoid turning this section into an unreadable
mess of proof trickery, we have omitted the formal proof that
the GWV axioms imply the Rushby axioms.

Similarly to the previous section, we prove that the axioms
B1 . . . B5 combined with the FW Blackens, FW Pol, and
Separation properties are sufficient to prove the desired
FW Correct theorem by reducing this problem to the re-
lated problem for Van der Meyden’s axiom. Doing so will
be the subject of the remainder of this paper.

5.3 Van der Meyden’s Axiom
Van der Meyden argues that these axioms defined by GWV
and Rushby unnecessarily restrict the class of systems to
which the results apply. He proposes an alternative formu-
lation consisting of one axiom over predicate black without
the need of ancillary functions, and thus the richness of the
state space they imply. Rushby’s axiom B5 basically states
that if the entire system is black, then this is still the case
in the next state. Van der Meyden generalises this notion
by requiring that if the value of a memory segment is com-
puted based only on the values of a set of memory segments
X, and all segments in X are black, then the computed seg-
ment must be black in the next state.

The notion of being computed based only on a set of memory
segments is just the same that GWV use to define the Sep-
aration axiom. In the Separation axiom, the requirement of

1Here the fact that the number of segments if finite is crit-
ical: without this requirement, the function blacken could
not be defined based on scrub.



the security policy is that the content of a memory segment
in the next state is a function of its current content, the
active partition, and the contents of the memory segments
that are allowed to influence it. Using the same construction,
Van der Meyden defines his sole axiom formalising black as
follows:

fun equals :: P(Segment)⇒ State⇒ State⇒ B where

equals(X, s, t) = ∀a ∈ X.select(s, a) = select(t, a)

assumes Black: ∀X ∈ P(Segment), s ∈ State,
a ∈ Segment.

(∀r, t ∈ Segment.equals(X, r, t) ∧
current(r) = current(t)→
select(next(r), a) = select(next(t), a)) ∧

(∀b ∈ X.black(s, b))→
black(next(s), a)

This axiom states that if the value of segment a in the next
state of s is a function of the segments in X and the active
partition, then this function preserves blackness. In other
words, as the value of a is computed based on X, so is the
blackness of a inherited from X.

The Black axiom follows from the Rushby axioms; indeed,
the proof for this in Isabelle is very simple. The proof starts
by assuming the two premises stated in the Black axiom:

fix X s a
assume 1:

(∀r, t ∈ Segment.equals(X, r, t) ∧
current(r) = current(t)→
select(next(r), a) = select(next(t), a))

assume (∀b ∈ X.black(s, b))

Because a is already black in s and is thus unaffected by
blacken due to B2, and because blacken does not change
the active partition of s, according to assumption 1 we have
select(next(s), a) = select(next(blacken(s)), a).

hence select(next(s), a) = select(next(blacken(s)), a)
by (metis 1 B2 B3 equals_def)

But of course blacken(s) is black for all segments; by B5, so
is next(blacken(s)). Because by B4 blackness is a function
of the contents of a memory segment, we get

thus black(next(s), a)
by (metis B1 B4 B5)

which completes the proof.

Unfortunately, the Black axiom is not sufficient to show
the FW Correct security requirement. This problem is
the topic of the next section.

6. PRESERVATION OF BLACKNESS
In the paper introducing the Black axiomatisation of the
black predicate[2], Van der Meyden appears to prove that
together with the FW Pol, FW Blackens, and Separa-
tion axioms, the Black axiom is sufficient to prove the
FW Correct theorem specifying the secure operation of
the Firewall. Unfortunately, this proof is incorrect, and the
Black axiom is in fact not strong enough to ensure that the
black predicate is sufficiently well-behaved.

To prove FW Correct, Van der Meyden correctly shows
that partitions other than B, including the Firewall parti-
tion F , can never make any segments of B nonblack. He

also shows that B can never make any segments other than
outbox nonblack without some other segment already be-
ing nonblack. A problem occurs, however, in proving that
B can never make outbox nonblack. For this to be the case
due to the Black axiom, there needs to be a set of segments
X such that the next contents of outbox is a function of
the active partition and the contents of the segments in X.

Van der Meyden shows that a set of segments X exists such
that among all states s for which current(s) = B, the
contents of outbox in next(s) is a function of the con-
tents of the segments in X. That is, he constructs a set X
such that for all segments r and t for which current(r) =
current(t) = B, if equals(X, r, t) holds, then we can con-
clude select(next(r),outbox) = select(next(t),outbox).
Based on the “no spontaneous generation of nonblack data”
intuition, we would expect this to be sufficient to show that
black(next(s),outbox) holds under the assumption that
current(s) = B and all segments in X are black in s,
and Van der Meyden argues exactly this in his proof of
FW Correct. This does not follow from the Black axiom,
however.

Indeed, there is nothing in the Black axiom that requires a
computation step of B to maintain the blackness of outbox
when all segments accessible to B are black. This would
require the next contents of outbox to be a function of the
active partition and segs(B). But in general, this is not
the case; the Firewall partition generally writes to outbox
based on segments not accessible to B, which means that
the next content of outbox is not independent of those
segments. Appendix A describes a specific counterexam-
ple in which this is the case, showing that the Separation,
FW Pol, FW Blackens, and Black axioms can all be true
while FW Correct is false.

To remedy this, we propose a stronger version of the Black
axiom that does not suffer from this problem, which we feel
better formalises the intuition behind the Black axiom. In
his flawed proof, Van der Meyden inadvertently argues that
the next contents of outbox are a function of the active
partition and the contents of a set X of segments among
those states s for which current(s) = B; because this pred-
icate current(s) = B is true for the specific state he is
considering, he concludes that blackness follows for outbox
in the next state of this specific state. We feel this line of
reasoning should hold for black for arbitrary predicates of
s. That is, if the functionality of segment a on segments
X property holds for all states matching some predicate P ,
and all segments of X are black in a state s also matching
this predicate P , then a should be black in the next state of
s. Formally:

assumes StrongBlack: ∀P ∈ P(State),
X ∈ P(Segment), s ∈ State, a ∈ Segment.

(∀r, t ∈ Segment.P (r) ∧ P (t) ∧
equals(X, r, t) ∧
current(r) = current(t)→
select(next(r), a) = select(next(t), a)) ∧

(∀b ∈ X.black(s, b)) ∧
P (s)→
black(next(s), a)

This definition is a generalisation of Black; using P (s) =
true yields Black as a special case. Using this stronger ax-



iom, Van der Meyden’s proof does hold; the problem de-
scribed above no longer applies, and the rest of the proof
goes through unchallenged.

We feel that the StrongBlack axiom is a more accurate
characterisation of the idea that nonblack data cannot be
generated from black data. The added predicate makes it
possible to show more fine-grained instances of subsystems
only having access to black data, and conclude the expected
consequences of that fact for these limited cases. In the
next section, we demonstrate how this notion can be used
to conclude useful properties of black.

Unlike the Black axiom, the StrongBlack axiom does not
follow from Rushby’s formalisation, and neither does the
Rushby formalisation follow from the StrongBlack axiom.
The two formalisations are formally incomparable. For both
directions, the reason that the implication does not hold is
straightforward. Rushby’s blacken function requires the
existence of a large set of states, including lots of states in
which all segments are black; the StrongBlack axiom, how-
ever, can easily hold in systems in which particular segments
are always black. Conversely, because the blacken function
does not preserve arbitrary predicates P , it is of no help in
proving the StrongBlack axiom for arbitrary values of P .
Constructing specific counterexamples for both implications
is left as an exercise for the reader.

Like Rushby’s axiomatisation, the StrongBlack axiom is
sufficient to prove the correctness of the Firewall when com-
bined with the Separation, FW Pol, and FW Blackens
postulates. We will prove both in the next section.

7. PROVING FW_CORRECT
In this section, we formally prove the correctness of the Fire-
wall under the GWV, Rushby, and StrongBlack axioma-
tisations of the black predicate. That is, for these three
axiomatisations, we prove that those axioms combined with
the Separation, FW Pol, and FW Blackens axioms to-
gether imply the security property FW Correct.

As described in Section 5.2, the Rushby axioms follow from
the GWV axioms. We can therefore prove the correctness of
both using a single proof that uses only the Rushby axioms
as an assumption. No such luck applies to the Rushby and
StrongBlack axiomatisations, however.

To avoid having to prove the same property twice for the
Rushby and StrongBlack axioms, we first construct an ax-
iomatisation that is weaker than either. This axiomatisation
only functions as an artefact of proof; it does not aim to fully
characterise the black predicate, but is only there to sim-
plify the proofs. The axiomatisation we have in mind is a
variant of StrongBlack that generalises Black in a min-
imal way while still being powerful enough to support the
attempted usage in Van der Meyden’s proof:

assumes WeakBlack: ∀X ∈ P(Segment), s ∈ State,
a ∈ Segment.

(∀r, t ∈ Segment.current(s) = current(r) ∧
equals(X, r, t) ∧
current(r) = current(t)→
select(next(r), a) = select(next(t), a)) ∧

(∀b ∈ X.black(s, b))→

black(next(s), a)

The WeakBlack axiom is a special case of the Strong-
Black axiom produced by substituting the predicate P (t) ≡
current(s) = current(t) for the variable P ; thus, it triv-
ially follows from StrongBlack. More interestingly, it also
follows from Rushby’s axioms, using almost exactly the same
proof as the one in Section 5.3:

fix X s a
assume 1:

(∀r, t ∈ Segment.
current(s) = current(t)∧current(r) = current(t)∧
equals(X, r, t)→
select(next(r), a) = select(next(t), a))

assume (∀b ∈ X.black(s, b))
hence select(next(s), a) = select(next(blacken(s)), a)
by (metis 1 B2 B3 equals_def)

thus black(next(s), a)
by (metis B1 B4 B5)

When using the WeakBlack axiom instead of Black, Van
der Meyden’s proof is correct. We prove this below by pre-
senting Van der Meyden’s proof in fully formalised form in
the Isabelle proof system.

The theorem we want to prove is that FW Correct holds
under the assumption of the Separation, FW Blackens,
FW Pol, and WeakBlack axioms:

theorem

assumes Separation: ∀s, t ∈ State, a ∈ Segment.
equals(dia(a) ∩ segs(current(s)), s, t) ∧

current(s) = current(t) ∧
select(s, a) = select(t, a)→

select(next(s), a) = select(next(t), a)

assumes FW Pol: ∀a, b ∈ Segment, P ∈ Partition.
a ∈ segs(B) ∧
b ∈ dia(a) ∧
b ∈ segs(P ) ∧
P 6= B →
(P = F ∧ a = outbox)

assumes FW Blackens: ∀s ∈ State.
current(s) = F ∧ black(s,outbox)→
black(next(s),outbox)

assumes WeakBlack: ∀X ∈ P(Segment), s ∈ State,
a ∈ Segment.

(∀r, t ∈ Segment.current(s) = current(r) ∧
equals(X, r, t) ∧
current(r) = current(t)→
select(next(r), a) = select(next(t), a)) ∧

(∀b ∈ X.black(s, b))→
black(next(s), a)

shows ∀s ∈ State, n ∈ N.
(∀a ∈ segs(B).black(s, a))→
(∀a ∈ segs(B).black(run(n, s), a))

Proof of this property is ultimately by induction on n. To
simplify things, we first prove the correctness of the induc-
tion step as a lemma.

proof -



have 0: ∀s ∈ State, a ∈ Segment.
(∀b ∈ segs(B).black(s, b))→
a ∈ segs(B)→ black(next(s), a)

proof -

fix s a
assume 1: ∀b ∈ segs(B).black(s, b)
assume 2: a ∈ segs(B)

We now need to prove that black(next(s), a). This proof
will proceed by cases, but first we prove the simple lemma
that FW Pol applies to a: in other words, that if something
can influence a, then a must be outbox and that something
must be accessible only to B and F . This is a triviality, but
proving it here will spare us the effort of having to duplicate
the proof in all the cases that make use of it.

with FW Pol have 3:

∀b ∈ Segment, P ∈ Partition.
b ∈ segs(P )→ P 6= B →
b ∈ dia(a)→ (a = outbox ∧ P = F )

by simp

The proof of black(next(s), a) will proceed by cases. We
first consider the case where a 6= outbox.

show black(next(s, a))
proof cases

assume 4: a 6= outbox

Because a 6= outbox, by FW Pol and Separation it fol-
lows that the next contents of a are a function of the con-
tents of segs(B) and the active partition. The WeakBlack
axiom then requires that the blackness of the segments in
segs(B) which we assumed in assumption 1. To show this,
we first need to establish the functional dependence of a on
segs(B) and the active partition as a lemma to later feed to
WeakBlack.

have ∀r, t ∈ State.
current(s) = current(r)→
current(r) = current(t)→
equals(segs(B), r, t)→
select(next(r), a) = select(next(t), a)

proof auto

fix r t
assume 5: current(s) = current(t)
assume 6: current(r) = current(t)
assume 7: equals(segs(B), r, t)

Because a ∈ segs(B) and equals(segs(B), r, t), we have
select(r, a) = select(t, a).

with 2 have 8: select(r, a) = select(t, a)
unfolding equals_def by simp

Because of FW Pol and the fact that a 6= outbox, we must
have dia(a) ⊆ segs(B). Because equals(segs(B), r, t) and
(X ∩ Y ) ⊆ X, we certainly have the following:

from 3 4 7 have

equals(dia(a) ∩ segs(current(r)), r, t)
unfolding equals_def by auto

But then Separation gives us the desired result:

with 6 8 Separation show

select(next(r), a) = select(next(t), a)
by simp

qed

This finishes the lemma stating that the next contents of a
are a function of the contents of segs(B) and the active par-
tition. The WeakBlack axiom will now prove the blackness
of a in next(s).

with 1 WeakBlack show black(next(s), a) by auto

This finishes the case where a 6= outbox.

Next, we make a further case distinction on the value of the
active partition in s. We consider three cases: current(s) =
B, current(s) = F , and current(s) 6= B∧current(s) 6= F .

next assume 4: a 6= outbox
show ?thesis proof cases

assume 5: current(s) = B

The case where current(s) = B uses the same structure as
the a 6= outbox case. It first proves the lemma that the
next contents of a are a function of the contents of segs(B)
and the active partition.

have ∀r, t ∈ State.
current(s) = current(r)→
current(r) = current(t)→
equals(segs(B), r, t)→
select(next(r), a) = select(next(t), a)

proof auto

fix r t
assume 6: current(s) = current(t)
assume 7: current(r) = current(t)
assume 8: equals(segs(B), r, t)
with 2 have 9: select(r, a) = select(t, a)
unfolding equals_def by simp

The details of this step are different from the a 6= outbox
case, however. Here, the next contents of a are only a func-
tion of the contents of segs(B) and the active partition
for those states s with current(s) = B; in other words,
this is the point in the proof in which the difference be-
tween WeakBlack and Black is crucial, and it’s the point
where Van der Meyden’s original proof is incorrect. Be-
cause current(r) = current(s), we have equals(dia(a) ∩
segs(current(r)), r, t):

from 5 6 7 8 have

equals(dia(a) ∩ segs(current(r)), r, t)
unfolding equals_def by simp

The remainder of this case proceeds in the same way as the
case where a 6= outbox.

with 7 9 Separation show

select(next(r), a) = select(next(t), a)
by simp

qed

with 1 WeakBlack show black(next(s), a) by auto

This concludes the case where current(s) = B.

The case where current(s) = F is almost trivial; blackness
of a in next(s) follows immediately from its blackness in s
and FW Blackens:

next assume 5: current(s) 6= B
show ?thesis proof cases

assume 6: current(s) = F
from 1 2 4 have black(s,outbox) by simp

with 4 6 FW Blackens show ?thesis by simp



All that remains is the case where both current(s) 6= B ∧
current(s) 6= F . This, too, is a simple case: the security
policy forbids the active partition from modifying any of
the segments of B, which means the same proof used for the
a 6= outbox case applies.

next assume 6: current(s) 6= F
have ∀r, t ∈ State.

current(s) = current(r)→
current(r) = current(t)→
equals(segs(B), r, t)→
select(next(r), a) = select(next(t), a)

proof auto

fix r t
assume 8: current(s) = current(t)
assume 9: current(r) = current(t)
assume equals(segs(B), r, t)
with 2 have 10: select(r, a) = select(t, a)
unfolding equals_def by simp

with 3 5 6 8 9 have

equals(dia(a) ∩ segs(current(r)), r, t)
unfolding equals_def by auto

with 9 10 Separation show

select(next(r), a) = select(next(t), a)
by simp

qed

with 1 WeakBlack show black(next(s), a) by auto

qed qed qed qed

Because this is the last case, this finishes the proof of the
lemma which states the correctness of the induction step
of the main theorem. That is, we just proved ∀s, a.(∀b ∈
segs(B).black(s, b))→ a ∈ segs(B)→ black(next(s), a).

With this lemma in hand, we can now easily prove the main
theorem, with an appeal to the lemma 0 in the induction
step:

shows ∀s ∈ State, n ∈ N.
(∀a ∈ segs(B).black(s, a))→
(∀a ∈ segs(B).black(run(n, s), a))

proof -

fix s n
assume ∀a ∈ segs(B).black(s, a)
then show ∀a ∈ segs(B).black(run(n, s), a)
proof (induction n, auto)

fix n x
assume 2: ∀x ∈ segs(B).black(run(n, s), x)
assume 3: x ∈ segs(B)
with 0 2 show black(next(run(n, s)), x) by simp

qed qed qed

This completes the proof.

8. CONCLUSION
In the previous section, we have proven the formal property
Separation∧FW Pol∧FW Blackens∧WeakBlack→
FW Correct. That is, we have shown that the desired se-
curity property that the untrusted application does not gain
access to unprivileged information holds, under assumptions
that the separation kernel and Firewall application behave
in a certain way.

The point of this exercise is to study techniques for the for-
mal verification of system properties in a compositional way.

In this paper we did not prove any properties about the be-
haviour of system components such as the separation ker-
nel or the Firewall; instead, these properties were taken for
granted. The problem studied in this paper is how to make
use of independently proven properties describing individ-
ual system components to prove properties of the system
as a whole in which these components play a role; that is,
to compose verified behaviour of components into verified
behaviour of the complete system.

When verifying practical systems, one would presumably in-
dependently prove behavioural properties regarding individ-
ual system components, and then later attempt the compo-
sition in a way similar to the methods used in this paper.
Based on our experience formalising the notions presented
in this paper, we feel confident that compositional formal
validation of system properties is a practical technique for
certifying desired system properties in applications such as
security certification.

Acknowledgments
We acknowledge funding from the European Union’s Sev-
enth Framework Programme (FP7/2007-2013) under grant
agreement no 318353 (EURO-MILS project: http://www.

euromils.eu).

9. REFERENCES
[1] J. Alves-Foss and C. Taylor. An analysis of the GWV

security policy. In In Fifth International Workshop on
ACL2 Prover and Its Applications, 2004.

[2] R. V. der Meyden. Remarks on the gwv firewall.
Available at http://www.cse.unsw.edu.au/~meyden/

research/gwv-firewall.pdf, October 2010.

[3] D. Greve. Information security modeling and analysis.
In D. S. Hardin, editor, Design and Verification of
Microprocessor Systems for High-Assurance
Applications, pages 249–299. Springer US, 2010.

[4] D. Greve, M. Wilding, R. Richards, and W. M.
Vanfleet. Formalizing security policies for dynamic
and distributed systems. Unpublished, Sept. 2004.

[5] D. Greve, M. Wilding, and W. M. Vanfleet. A
separation kernel formal security policy. In Fourth
International Workshop on the ACL2 Theorem Prover
and Its Applications (ACL2 ’03), July 2003.

[6] M. Kaufmann, P. Manolios, and J S. Moore. ACL2
Computer-Aided Reasoning: An Approach, 2000.

[7] T. Nipkow, L. Paulson, and M. Wenzel. Isabelle/HOL:
A Proof Assistant for Higher-Order Logic, volume
2283 of LNCS. 2002.

[8] S. Owre, J. Rushby, and N. Shankar. PVS: A
Prototype Verification System. In Proceedings of the
Eleventh International Conference on Automated
Deduction (CADE’92), volume 607, pages 748–752,
June 1992.

[9] J. Rushby. A separation kernel formal security policy
in PVS. Technical report, Computer Science
Laboratory, SRI international, 2004.

[10] M. W. Whalen, D. A. Greve, and L. G. Wagner.
Model checking information flow. In D. S. Hardin,
editor, Design and Verification of Microprocessor
Systems for High-Assurance Applications, pages
381–428. Springer US, 2010.



APPENDIX
A. COUNTEREXAMPLE TO

VAN DER MEYDEN’S AXIOM
In Section 6, we described how a system can be constructed
that satisfies the Separation, FW Pol, FW Blackens,
and Black axioms, while still not satisfying FW Correct.
For the sake of completion, we provide here a specific mini-
mal system for which this is the case.

Consider a GWV system as described in Section 3 con-
sisting of two segments outbox and inbox, a partition B
with segs(B) = {outbox}, and a second partition F with
segs(F ) = {outbox, inbox}. The dia function does not
constrain any types of influence: dia(a) = {outbox, inbox}
for both values of a. The system has three possible states,
named S1, S2, and S3. The three states succeed each other
in a cycle: next(S1) = S2, next(S2) = S3, and next(S3) =
S1.

The contents of the memory, its sensitivity, and the active
partition are summarized in Table 1. The F partition is
active in the states S1 and S2, and B is active in S3. The
contents of outbox are equal for states S1 and S2, and
different for S3; the contents of inbox are equal for S1 and
S3 and differnet for S2. Of course, the exact values are
irrelevant.

The outbox segment is black in states S2 and S3; the inbox
segment is never black. This has the curious property that
the states S1 and S2 have the same contents for the outbox
segment, but differing blackness for that segment; the GWV
and Rushby axiomatisations of the black predicate would
not allow this, but it is possible under the Black axiom.

The system described here satisfies the Separation axiom.
Because the dia function describes the complete relation and
thus does not forbid anything, and because B does not write
to inbox in the one state in which it is active, this is trivial.

Because B and F are the only partitions in the system and
outbox is the only segment in segs(B), FW Pol is triv-
ially satisfied. The FW Blackens axiom is easily checked
by verifying that black(next(s),outbox) is true for every
state s with current(s) = F .

Showing that this system satisfies Black is less obvious. The
value of outbox in the next state of s is not a function of the
current value of {outbox} and the active partition; indeed,
the states S1 and S2 have the same active partition and the
same values for outbox, yet select(next(S1),outbox) =
1 6= 2 = select(next(S2),outbox). The value of outbox
in the next stage of s is a function of the current value of
{inbox} and the active partition, and therefore also of the
superset {outbox, inbox}. However, all we can conclude
from this using the Black axiom is that if all segments in

{inbox} are black, then outbox must be black in the next
state. Because inbox is never black, this is vacuously true.
Thus, the system satisfies Black in a vacuous way.

The WeakBlack axiom, and therefore also the Strong-
Black axiom, would note that among all states for which
current(s) = B, the contents of outbox in the next state
is a function of the active partition and the contents of
{outbox}; indeed, it is even a function of the active parti-
tion and the contents of ∅. Thus, they would require that
whenever outbox is black in a state for which B is the ac-
tive partition, then outbox must still be black in the next
state. This system, then, does not satisfy the WeakBlack
axiom, and is not a counterexample against it.

For the Black axiom, however, it is indeed a counterexam-
ple. For this system does not satisfy FW Correct; in state
S3 all segments of B are black, yet in the next state S1,
this is no longer the case. That makes this a system for the
Separation, FW Pol, FW Blackens, and Black axioms
are satisfied, yet FW Correct does not hold, and a proof
that Van der Meyden’s proof is flawed.

s next(s) current(s) select(s,outbox) select(s, inbox) black(s,outbox) black(s, inbox)
S1 S2 F 1 3 false false
S2 S3 F 1 4 true false
S3 S1 B 2 3 true false

Table 1: The Firewall MILS Example.


